ΒΑΣΙΚΕΣ ΜΕΘΟΔΟΛΟΓΙΕΣ

Άρτιοι - Περιττοί

```
ΔΙΑΒΑΣΕ Χ
ΑΝ Χ MOD 2 = 0 ΤΟΤΕ
    ΓΡΑΨΕ 'ο Χ είναι άρτιος'
ΑΛΛΙΩΣ
    ΓΡΑΨΕ 'ο Χ είναι περιττός'
ΤΕΛΟΣ_ΑΝ
```

Πολλαπλάσια του Ν

```
ΔΙΑΒΑΣΕ Χ
ΑΝ Χ MOD Ν = 0 ΤΟΤΕ
    ΓΡΑΨΕ 'ο Χ είναι πολλαπλάσιο του', Ν
ΑΛΛΙΩΣ
    ΓΡΑΨΕ 'ο Χ ΔΕΝ είναι πολλαπλάσιο του', Ν
ΤΕΛΟΣ_ΑΝ
```

Διαχείριση ψηφίων ακεραίου

Αν **x** ο ελεγχόμενος αριθμός, τότε με τις εντολές **x1 ← x MOD 10 ^ ν** και **x2 ← x DIV 10 ^ ν** , στη μεταβλητή x1 εκχωρούμε τα τελευταία ν ψηφία του, και στη μεταβλητή x2 τα προηγούμενα από αυτά.
Αν π.χ. είναι x = 12345, τότε μετά την εντολή **x1 ← x MOD 10 ^ 3** θα είναι x1 = 345 (τα τελευταία 3 ψηφία), ενώ μετά την **x2 ← x DIV 10 ^ 3** θα είναι x2 = 12 (τα προηγούμενα από τα 3 τελευταία ψηφία).

Να διαβαστεί τετραψήφιος ακέραιος, μετά τα δύο πρώτα ψηφία του να παρεμβληθεί το 9 και να εκτυπωθεί ο πενταψήφιος που προέκυψε.

```
ΔΙΑΒΑΣΕ x               ! έστω ότι δίνεται ο τετραψήφιος 1234
x1 ← x MOD 10 ^ 2       ! x1 = 34
x2 ← x DIV 10 ^ 2       ! x2 = 12
x2 ← x2 * 10 ^ 3        ! x2 = 12000
x3 ← 9 * 10 ^ 2         ! το 9 θα σημαίνει εκατοντάδες, οπότε x3 = 900
x ← x2 + x3 + x1        ! x = 12000 + 900 + 34 = 12934
ΓΡΑΨΕ x
```

Να διαβαστεί εξαψήφιος ακέραιος και να εκτυπωθούν τα ψηφία του από το τελευταίο προς το πρώτο.

```
ΔΙΑΒΑΣΕ x                    ! έστω ότι δίνεται ο τετραψήφιος 123456
y ← 0
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 6
    x1 ← x MOD 10            ! x1 = 6 κ.ο.κ. (το τελευταίο ψηφίο)
    y ← y + x1 * 10 ^ (7 – i) ! y = 600.000 κ.ο.κ. (γίνεται πρώτο)
    x ← x DIV 10             ! x = 12345 κ.ο.κ. (τα υπόλοιπα ψηφία)
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ y                      ! y = 654321
```

Απαλοιφή των ΚΑΙ, Ή στη συνθήκη της ΑΝ

Το **ΚΑΙ** στην έκφραση **ΑΝ** συνθήκη1 **ΚΑΙ** συνθήκη2 **ΤΟΤΕ**, απαλείφεται με δύο **ΑΝ**, όπου η συνθήκη της $1^{ης}$ είναι η συνθήκη1, η συνθήκη της $2^{ης}$ είναι η συνθήκη2 και η $2^{η}$ **ΑΝ** είναι εμφωλευμένη στην $1^{η}$.

Να γραφεί η:

```
ΑΝ (x = 2  ΚΑΙ  y <> 3) ΤΟΤΕ
      ΕΜΦΑΝΙΣΕ 'ΟΚ'
ΤΕΛΟΣ_ΑΝ
```

χωρίς το ΚΑΙ.

```
ΑΝ x = 2 ΤΟΤΕ
   ΑΝ y <> 3 ΤΟΤΕ
      ΕΜΦΑΝΙΣΕ 'ΟΚ'
   ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΑΝ
```

Το **Ή** στην έκφραση **ΑΝ** συνθήκη1 **Ή** συνθήκη2 **ΤΟΤΕ**, απαλείφεται ως εξής:

```
   ΑΝ συνθήκη1 ΤΟΤΕ
      εντολές
   ΑΛΛΙΩΣ_ΑΝ συνθήκη2 ΤΟΤΕ
      οι ίδιες εντολές
   ΤΕΛΟΣ_ΑΝ
```

Να γραφεί η:

```
ΑΝ (z = 2  Ή  w <> 3) ΤΟΤΕ
      ΕΜΦΑΝΙΣΕ 'ΟΚ'
ΤΕΛΟΣ_ΑΝ
```

χωρίς το Ή.

```
ΑΝ z = 2 ΤΟΤΕ
   ΕΜΦΑΝΙΣΕ 'ΟΚ'
ΑΛΛΙΩΣ_ΑΝ w <> 3 ΤΟΤΕ
   ΕΜΦΑΝΙΣΕ 'ΟΚ'
ΤΕΛΟΣ_ΑΝ
```

Να γραφεί η:

```
ΑΝ (x = 2  ΚΑΙ  y <> 3)  ΚΑΙ  (z = 2  Ή  w <> 3) ΤΟΤΕ
      ΕΜΦΑΝΙΣΕ 'ΟΚ'
ΤΕΛΟΣ_ΑΝ
```

χωρίς τους λογικούς τελεστές.

```
ΑΝ x = 2 ΤΟΤΕ
   ΑΝ y <> 3 ΤΟΤΕ
      ΑΝ z = 2 ΤΟΤΕ
         ΕΜΦΑΝΙΣΕ 'ΟΚ'
      ΑΛΛΙΩΣ_ΑΝ w <> 3 ΤΟΤΕ
         ΕΜΦΑΝΙΣΕ 'ΟΚ'
      ΤΕΛΟΣ_ΑΝ
   ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΑΝ
```

Μέσος Όρος

Αθροίζουμε όλα τα στοιχεία με τη βοήθεια ενός αθροιστή π.χ. S που αρχικοποιείται με την τιμή μηδέν και στον οποίο προστίθενται διαδοχικά όλες οι τιμές.
Σχετικά με το πλήθος, εάν δεν είναι γνωστό εκ των προτέρων, χρησιμοποιούμε μετρητή, π.χ. C, που αρχικοποιείται με μηδέν και αυξάνει κατά ένα για κάθε καινούργιο στοιχείο.
Εάν οι τιμές προστίθενται μετά από έλεγχο κάποιας συνθήκης, πρέπει πριν υπολογίσουμε το μέσο όρο ως S / C να ελέγξουμε εάν C <> 0.

Να διαβαστούν 50 αριθμοί και να εκτυπωθεί ο μέσος όρος τους.

```
S ← 0                          ! αρχικοποίηση του αθροιστή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 50
    ΔΙΑΒΑΣΕ X
    S ← S + X                  ! ενημέρωση του αθροιστή
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
MO ← S / 50                    ! το πλήθος είναι γνωστό
ΓΡΑΨΕ MO
```

Να διαβαστούν 50 ακέραιοι αριθμοί και να εκτυπωθεί ο μέσος όρος των άρτιων.

```
S ← 0                          ! αρχικοποίηση του αθροιστή
C ← 0                          ! αρχικοποίηση του μετρητή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 50
    ΔΙΑΒΑΣΕ X
    ΑΝ X MOD 2 = 0 ΤΟΤΕ        ! εάν είναι άρτιος…
        S ← S + X              ! …ενημερώνουμε τον αθροιστή…
        C ← C + 1              ! …και τον μετρητή
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΑΝ C <> 0 ΤΟΤΕ                 ! εάν υπήρξαν άρτιοι…
    MO ← S / C                 ! …υπολογίζουμε το μέσο όρο
    ΓΡΑΨΕ MO
ΤΕΛΟΣ_ΑΝ
```

Να διαβαστούν επαναληπτικά αριθμοί, μέχρι να δοθεί το 0 και να εκτυπωθεί ο μέσος όρος τους.

```
S ← 0                          ! αρχικοποίηση του αθροιστή
C ← 0                          ! αρχικοποίηση του μετρητή
ΔΙΑΒΑΣΕ X
ΟΣΟ X<>0 ΕΠΑΝΑΛΑΒΕ
    S ← S + X                  ! …ενημερώνουμε τον αθροιστή…
    C ← C + 1                  ! …και τον μετρητή
    ΔΙΑΒΑΣΕ X
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΑΝ C <> 0 ΤΟΤΕ                 ! εάν υπήρξαν μη μηδενικοί αριθμοί…
    MO ← S / C                 ! …υπολογίζουμε το μέσο όρο
    ΓΡΑΨΕ MO
ΤΕΛΟΣ_ΑΝ
```

Ποσοστά

Υπολογίζουμε το πλήθος των στοιχείων, των οποίων μας ενδιαφέρει το ποσοστό, χρησιμοποιώντας μετρητή.
Μετά διαιρούμε με το σύνολο, αφού ελέγξουμε πως δεν είναι 0 και πολλαπλασιάζουμε επί 100.

Να διαβαστούν 50 ακέραιοι αριθμοί και να εκτυπωθεί το ποσοστό των άρτιων.

```
C ← 0                                ! αρχικοποίηση του μετρητή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 50
    ΔΙΑΒΑΣΕ X
    ΑΝ X MOD 2 = 0 ΤΟΤΕ              ! εάν είναι άρτιος…
        C ← C + 1                    ! … ενημερώνουμε τον μετρητή
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ποσοστό ← (C / 50) * 100             ! υπολογίζουμε το ποσοστό
ΓΡΑΨΕ ποσοστό
```

Να διαβαστούν επαναληπτικά ακέραιοι αριθμοί, μέχρι να δοθεί το 0 και να εκτυπωθεί το ποσοστό των άρτιων μεταξύ των αρνητικών.

```
C1 ← 0                               ! αρχικοποίηση του μετρητή αρνητικών
C2 ← 0                               ! αρχικοποίηση του μετρητή άρτιων
ΔΙΑΒΑΣΕ X
ΟΣΟ X<>0 ΕΠΑΝΑΛΑΒΕ
    ΑΝ X < 0 ΤΟΤΕ                    ! εάν είναι αρνητικός…
        C1 ← C1 + 1                  ! … ενημερώνουμε τον μετρητή των αρνητικών
        ΑΝ X MOD 2 = 0 ΤΟΤΕ          ! εάν είναι άρτιος αρνητικός…
            C2 ← C2 + 1              ! … ενημερώνουμε τον μετρητή των άρτιων
        ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΑΝ
    ΔΙΑΒΑΣΕ X
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΑΝ C1 <> 0 ΤΟΤΕ                      ! εάν υπήρξαν αρνητικοί αριθμοί…
    ποσοστό ← (C2 / C1) * 100        ! υπολογίζουμε το ποσοστό τους
    ΓΡΑΨΕ ποσοστό
ΤΕΛΟΣ_ΑΝ
```

Έλεγχος ακεραίων

Για να ελέγξουμε εάν ένας αριθμός είναι ή όχι ακέραιος, τον συγκρίνουμε με το ακέραιο μέρος του.

```
ΔΙΑΒΑΣΕ X
ΑΝ X = Α_Μ(X) ΤΟΤΕ
    ΓΡΑΨΕ 'ο X είναι ακέραιος'
ΑΛΛΙΩΣ
    ΓΡΑΨΕ 'ο x ΔΕΝ είναι ακέραιος'
ΤΕΛΟΣ_ΑΝ
```

Μέγιστα - Ελάχιστα

Η αρχικοποίηση του **min** πρέπει να γίνεται είτε με μία υπαρκτή τιμή είτε με μία τιμή που να ξεπερνά τη μεγαλύτερη δυνατή (εάν π.χ. πρόκειται για βαθμολογίες από 0 έως και 20, αρχικοποιούμε με min ← 21.
Η αρχικοποίηση του **max** πρέπει να γίνεται είτε με μία υπαρκτή τιμή είτε με μία τιμή που να υπολείπεται της μικρότερης δυνατής (εάν π.χ. πρόκειται για θερμοκρασίες από –20 έως και +30, αρχικοποιούμε με max ← –21.

Να διαβαστούν τρεις αριθμοί και να εκτυπωθεί ο μεγαλύτερος από αυτούς.

```
ΔΙΑΒΑΣΕ α, β, γ          ! έστω ότι α=1, β=2, γ=3
max ← α                  ! max=1 (αρχικοποίηση μέσω υπαρκτής τιμής)
ΑΝ β > max ΤΟΤΕ          ! 2 > 1 ΑΛΗΘΕΣ
    max ← β              ! max=2
ΤΕΛΟΣ_ΑΝ
ΑΝ γ > max ΤΟΤΕ          ! 3 > 2 ΑΛΗΘΕΣ
    max ← γ              ! max=3
ΤΕΛΟΣ_ΑΝ
ΓΡΑΨΕ max                ! εκτυπώνει 3
```

Να διαβαστούν επαναληπτικά 100 αριθμοί και να εκτυπωθεί ο μικρότερος από αυτούς.

```
ΔΙΑΒΑΣΕ α                ! για να μπορέσουμε να κάνουμε…
min ← α                  ! …αρχικοποίηση του ελαχίστου
ΓΙΑ i ΑΠΟ 2 ΜΕΧΡΙ 100    ! διαβάζουμε και τις υπόλοιπες τιμές
    ΔΙΑΒΑΣΕ α
    ΑΝ α < min ΤΟΤΕ      ! εάν η τρέχουσα τιμή είναι μικρότερη
        min ← α          ! τότε αυτή γίνεται το ελάχιστο
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ min
```

ΘΕΣΗ ΕΛΑΧΙΣΤΟΥ: Να διαβαστούν επαναληπτικά οι βαθμοί και τα ονόματα 100 μαθητών και να εκτυπωθεί η θέση του μικρότερου βαθμού (ΜΟΝΑΔΙΚΟΣ) και το αντίστοιχο όνομα.

```
ΔΙΑΒΑΣΕ βαθμός, όνομα    ! για να μπορέσουμε να κάνουμε…
min ← βαθμός             ! …αρχικοποίηση του ελαχίστου
pos ← 1                  ! αρχικοποίηση θέσης
name_min ← όνομα         ! αρχικοποίηση ονόματος με ελάχιστο βαθμό
ΓΙΑ i ΑΠΟ 2 ΜΕΧΡΙ 100    ! διαβάζουμε και τις υπόλοιπες τιμές
    ΔΙΑΒΑΣΕ βαθμός, όνομα
    ΑΝ βαθμός < min ΤΟΤΕ ! εάν ο νέος βαθμός είναι μικρότερος...
        min ← βαθμός     ! ...τότε αυτός γίνεται ο ελάχιστος...
        pos ← i          ! ...και κρατάμε τη θέση του...
        name_min ← όνομα ! ...και το αντίστοιχο όνομα
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ pos, name_min
```

Επανάληψη μέσω αθροιστικής συνθήκης

Πρόκειται για περιπτώσεις όπου δεν έχουμε συγκεκριμένο πλήθος επαναλήψεων, καθώς η συνθήκη τερματισμού είναι μία αξία η οποία μεταβάλλεται αθροιστικά (συν ή πλην) σε κάθε επανάληψη. (π.χ. αγορές μέχρι κάποιου ποσού).

Να διαβαστούν επαναληπτικά αριθμοί, εφόσον το άθροισμα τους δεν ξεπερνά το 1000 και να υπολογιστεί και να εμφανιστεί ο μέσος όρος τους, καθώς και το πόσο υπολειπόμαστε του 1000.

```
S ← 0
C ← 0
ΔΙΑΒΑΣΕ X

ΟΣΟ S + X <= 1000 ΕΠΑΝΑΛΑΒΕ          ! πρώτα ο έλεγχος…
    S ← S + X                        ! …και  έπειτα η άθροιση
    C ← C + 1
    ΔΙΑΒΑΣΕ X
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΑΝ C <> 0 ΤΟΤΕ              ! εάν υπήρξαν δεκτοί αριθμοί… (ή ισοδύναμα S<>0)
    ΜΟ ←  S / C             ! υπολογίζουμε το μέσο όρο
    ΓΡΑΨΕ ΜΟ
ΤΕΛΟΣ_ΑΝ

υπόλοιπο ← 1000 – S
ΓΡΑΨΕ υπόλοιπο
```

Να διαβαστούν επαναληπτικά αριθμοί, μέχρι το άθροισμα τους να ξεπεράσει το 1000 και να υπολογιστεί και να εμφανιστεί ο μέσος όρος τους, καθώς και το πόσο υπερβήκαμε το 1000.

```
S ← 0
C ← 0

ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
    ΔΙΑΒΑΣΕ X
    S ← S + X
    C ← C + 1
ΜΕΧΡΙΣ_ΟΤΟΥ (S > 1000)

ΜΟ ← S / C          ! σίγουρα δόθηκε μία τουλάχιστον τιμή
ΓΡΑΨΕ ΜΟ

επιπλέον ← S – 1000
ΓΡΑΨΕ επιπλέον
```

Κλιμακωτή επεξεργασία – Επικάλυψη διατημάτων

Όταν η τιμή της μεταβλητής βρεθεί σε κάποιο από τα διαστήματα, τότε υπολογίζεται ολόκληρο κάθε ένα από τα προηγούμενα διαστήματα επί τον αντίστοιχο συντελεστή και η διαφορά του αριστερού άκρου του τρέχοντος διαστήματος από την τιμή της μεταβλητής, επί τον αντίστοιχο συντελεστή.

Να υπολογιστεί και να εκτυπωθεί το ποσό της κλιμακωτής χρέωσης με βάση τον παρακάτω πίνακα.

χρόνος σε δευτερόλεπτα	χρέωση ανά δευτερόλεπτο
από 0 μέχρι και 100	0.01
πάνω από 100 μέχρι και 300	0.02
πάνω από 300	0.03

```
ΔΙΑΒΑΣΕ Χ              ! θεωρώντας πως το Χ θα είναι μη αρνητικός
ΑΝ Χ <= 100 ΤΟΤΕ
    χρέωση ← Χ * 0.01
ΑΛΛΙΩΣ_ΑΝ Χ <= 300 ΤΟΤΕ
    χρέωση ← 100 * 0.01 + (Χ - 100) * 0.02
ΑΛΛΙΩΣ
    χρέωση ← 100 * 0.01 + 200 * 0.02 + (Χ - 300) * 0.03
ΤΕΛΟΣ_ΑΝ
ΕΚΤΥΠΩΣΕ χρέωση
```

ΠΡΟΣΟΧΗ εάν το Χ μπορεί να πάρει και αρνητικές τιμές, το με το να γράψω:

```
ΔΙΑΒΑΣΕ Χ              ! το Χ μπορεί να είναι οποιοσδήποτε αριθμός
ΑΝ Χ >=0 ΚΑΙ Χ <= 100 ΤΟΤΕ
    χρέωση ← Χ * 0.01
ΑΛΛΙΩΣ_ΑΝ Χ <= 300 ΤΟΤΕ
    χρέωση ← 100 * 0.01 + (Χ - 100) * 0.02
. . .
```

ΔΕΝ ΕΙΜΑΙ ΚΑΛΥΜΕΝΟΣ

Αν για παράδειγμα δοθεί ως Χ το – 1, τότε η **ΑΝ** θα είναι Ψευδής, αλλά ή **ΑΛΛΙΩΣ_ΑΝ** θα εκτιμηθεί ως Αληθής και θα οδηγηθώ σε λάθος.

Ένας τρόπος για να αποφύγω το λάθος είναι:

```
    ΔΙΑΒΑΣΕ Χ              ! το Χ μπορεί να είναι οποιοσδήποτε αριθμός
    ΑΝ Χ < 0 ΤΟΤΕ          ! έλεγχος μη αποδεκτής τιμής
        ΓΡΑΨΕ "μη αποδεκτή τιμή"
    ΑΛΛΙΩΣ_ΑΝ Χ <= 100 ΤΟΤΕ
        χρέωση ← Χ * 0.01
    ΑΛΛΙΩΣ_ΑΝ Χ <= 300 ΤΟΤΕ
        χρέωση ← 100 * 0.01 + (Χ - 100) * 0.02
    ΑΛΛΙΩΣ
        χρέωση ← 100 * 0.01 + 200 * 0.02 + (Χ - 300) * 0.03
    ΤΕΛΟΣ_ΑΝ
ΕΚΤΥΠΩΣΕ χρέωση
```

Έλεγχος εγκυρότητας

Όταν θέλουμε να διαβάσουμε τιμές που να ικανοποιούν συγκεκριμένα κριτήρια, η εντολή **ΔΙΑΒΑΣΕ** τοποθετείται σε μία δομή επανάληψης **ΜΕΧΡΙ_ΟΤΟΥ** με συνθήκη τα κριτήρια που θέλουμε να ικανοποιούνται.

Να διαβαστεί μεταβλητή χαρακτήρας που να ξεκινά από το γράμμα 'Σ'

```
ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
    ΔΙΑΒΑΣΕ X
ΜΕΧΡΙΣ_ΟΤΟΥ (X >= 'Σ') ΚΑΙ (X < 'Τ')
```

Να διαβαστούν τα αποτελέσματα 16 αγώνων, με επιτρεπτές τιμές 'Ν', 'Ι' και 'Η' (από το Νίκη, Ισοπαλία και Ήττα αντίστοιχα).

```
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 16
    ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ                          ! έλεγχος εγκυρότητας
        ΔΙΑΒΑΣΕ ΑΠ
    ΜΕΧΡΙΣ_ΟΤΟΥ (ΑΠ = 'Ν') Ή (ΑΠ = 'Ι') Ή (ΑΠ = 'Η')
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Να διαβαστούν 100 ονόματα μαθητών και οι αντίστοιχοι βαθμοί τους, στην κλίμακα 0 έως 20 και να υπολογιστεί ο μέσος όρος τους.

```
S ← 0                                        ! αρχικοποίηση του αθροιστή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 100
    ΔΙΑΒΑΣΕ όνομα                            ! χωρίς έλεγχο
    ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ                          ! έλεγχος εγκυρότητας
        ΔΙΑΒΑΣΕ βαθμός
    ΜΕΧΡΙΣ_ΟΤΟΥ (βαθμός >= 0) ΚΑΙ (βαθμός <= 20)
    S_←  S + X                               ! ενημέρωση του αθροιστή
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΜΟ_←  S / 100                                ! το πλήθος είναι γνωστό
ΓΡΑΨΕ ΜΟ
```

Πολλαπλασιασμός αλά Ρωσικά

Διαβάζονται δύο ακέραιοι αριθμοί Μ1 και Μ2 και υπολογίζεται το γινόμενό τους Ρ.

```
ΔΙΑΒΑΣΕ Μ1, Μ2
Ρ ← 0
ΟΣΟ Μ2 > 0 ΕΠΑΝΑΛΑΒΕ
    ΑΝ Μ2 MOD  2 = 1 ΤΟΤΕ
        Ρ ← Ρ + Μ1
    ΤΕΛΟΣ_ΑΝ
    Μ1 ← Μ1 * 2                 ! ο Μ1 διπλασιάζεται (αριστερή ολίσθηση)
    Μ2 ← Μ2 DIV 2               ! ο Μ2 υποδιπλασιάζεται (δεξιά ολίσθηση)
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ Ρ
```

Μετατροπές

Η μετατροπή από **ΟΣΟ** σε **ΜΕΧΡΙΣ_ΟΤΟΥ** και το αντίθετο, γίνεται με αντιστροφή της συνθήκης.

```
i ← 1
ΟΣΟ (i <= 4) ΕΠΑΝΑΛΑΒΕ
    ΕΚΤΥΠΩΣΕ i
    i ← 2 * i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
i ← 1
ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
    ΕΚΤΥΠΩΣΕ i
    i ← 2 * i
ΜΕΧΡΙΣ_ΟΤΟΥ (i > 4)
```

Εάν η **ΟΣΟ** ενδέχεται να μην εκτελεστεί τουλάχιστον μία φορά, πρέπει οι εντολές της να γραφούν και πριν από αυτήν.

```
X ← A
ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
    X ← X + 1
    Y ← 10 – X
ΜΕΧΡΙΣ_ΟΤΟΥ (X >= 10)
ΕΚΤΥΠΩΣΕ X, Y
```

⇨

```
X ← A
X ← X + 1
Y ← 10 – X
ΟΣΟ (X < 10) ΕΠΑΝΑΛΑΒΕ
    X ← X + 1
    Y ← 10 – X
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΕΚΤΥΠΩΣΕ X, Y
```

Εάν η **ΜΕΧΡΙΣ_ΟΤΟΥ** ενδέχεται να εκτελεστεί, ενώ δεν πρέπει, πρέπει οι εντολές της να μπουν μέσα σε **ΑΝ** με συνθήκη τη συνθήκη της **ΟΣΟ**.

```
ΔΙΑΒΑΣΕ X
ΟΣΟ (X <> 0) ΕΠΑΝΑΛΑΒΕ
    ΕΚΤΥΠΩΣΕ X
    ΔΙΑΒΑΣΕ X
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
ΔΙΑΒΑΣΕ X
ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
    ΑΝ (X <> 0) τότε
        ΕΚΤΥΠΩΣΕ X
        ΔΙΑΒΑΣΕ X
    ΤΕΛΟΣ_ΑΝ
ΜΕΧΡΙΣ_ΟΤΟΥ (X = 0)
```

Κατά την μετατροπή της **ΓΙΑ** i **ΑΠΟ** τ1 **ΜΕΧΡΙ** τ2 **ΜΕ ΒΗΜΑ** β στην **ΟΣΟ** :

- Το i αρχικοποιείται πριν το βρόχο ως i ← τ1
- Η συνθήκη είναι **ΟΣΟ** (i <= τ2) εάν β > 0 και **ΟΣΟ** (i >= τ2) εάν β < 0
- Στο τέλος του βρόχου η i τροποποιείται ως εξής: i ← i + β

```
ΓΙΑ i ΑΠΟ 5 ΜΕΧΡΙ 1 ΜΕ ΒΗΜΑ –1
    ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
i ← 5
ΟΣΟ (i >= 1) ΕΠΑΝΑΛΑΒΕ
    ΕΚΤΥΠΩΣΕ i
    i ← i – 1
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Κατά την μετατροπή της **ΟΣΟ** σε **ΓΙΑ** i **ΑΠΟ** τ1 **ΜΕΧΡΙ** τ2 **ΜΕ ΒΗΜΑ** β (όταν αυτό είναι εφικτό):

- Η μεταβλητή της συνθήκης της **ΟΣΟ** γίνεται μεταβλητή της **ΓΙΑ**
- Η τιμή **ΑΠΟ** είναι η αρχική τιμή της μεταβλητής–μετρητή της **ΟΣΟ**
- Η τιμή **ΜΕΧΡΙ** είναι η τιμή στη συνθήκη της **ΟΣΟ**
- Η τιμή **ΜΕ ΒΗΜΑ** είναι το ποσό αύξησης ή μείωσης της μεταβλητής–μετρητή μέσα στο βρόχο της **ΟΣΟ**.

```
i ← 3
ΟΣΟ (i <= 9) ΕΠΑΝΑΛΑΒΕ
      ΕΚΤΥΠΩΣΕ i
      i ← i + 2
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
ΓΙΑ i ΑΠΟ 3 ΜΕΧΡΙ 9 ΜΕ ΒΗΜΑ 2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

```
i ← 15
ΟΣΟ (i >= 9) ΕΠΑΝΑΛΑΒΕ
      ΕΚΤΥΠΩΣΕ i
      i ← i – 2
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
ΓΙΑ i ΑΠΟ 15 ΜΕΧΡΙ 9 ΜΕ ΒΗΜΑ –2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Εάν από τη συνθήκη της **ΟΣΟ** λείπει η ισότητα (καθαρή ανίσωση), δημιουργούμε εμείς την ισότητα, μεταβάλλοντας κατάλληλα την τιμή της συνθήκης, λαμβάνοντας υπ' όψιν τις τιμές που δημιουργούνται βάση της αρχικής τιμής και του βήματος.

```
i ← 3
ΟΣΟ (i < 9) ΕΠΑΝΑΛΑΒΕ
      ΕΚΤΥΠΩΣΕ i
      i ← i + 2
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
i ← 3
ΟΣΟ (i < = 7) ΕΠΑΝΑΛΑΒΕ
      ΕΚΤΥΠΩΣΕ i
      i ← i + 2
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
ΓΙΑ i ΑΠΟ 3 ΜΕΧΡΙ 7 ΜΕ ΒΗΜΑ 2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

```
i ← 10
ΟΣΟ (i > 3) ΕΠΑΝΑΛΑΒΕ
      ΕΚΤΥΠΩΣΕ i
      i ← i – 2
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
i ← 10
ΟΣΟ (i > = 4) ΕΠΑΝΑΛΑΒΕ
      ΕΚΤΥΠΩΣΕ i
      i ← i – 2
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
ΓΙΑ i ΑΠΟ 10 ΜΕΧΡΙ 4 ΜΕ ΒΗΜΑ –2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Εάν η μεταβολή του μετρητή της **ΟΣΟ** γίνεται πριν από τις εντολές της (και όχι στο τέλος), τότε οι τιμές **ΑΠΟ** και **ΜΕΧΡΙ** μέσα στη **ΓΙΑ** θα πρέπει να μεταβληθούν κατά το **ΒΗΜΑ**

```
i ← 3
ΟΣΟ (i <= 9) ΕΠΑΝΑΛΑΒΕ
      i ← i + 2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
ΓΙΑ i ΑΠΟ 5 ΜΕΧΡΙ 11 ΜΕ ΒΗΜΑ 2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

```
i ← 10
ΟΣΟ (i > 3) ΕΠΑΝΑΛΑΒΕ
      i ← i – 2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
i ← 10
ΟΣΟ (i > = 4) ΕΠΑΝΑΛΑΒΕ
      i ← i – 2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

⇨

```
ΓΙΑ i ΑΠΟ 8 ΜΕΧΡΙ 2 ΜΕ ΒΗΜΑ –2
      ΕΚΤΥΠΩΣΕ i
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

ΜΕΘΟΔΟΛΟΓΙΕΣ ΣΤΟΥΣ ΠΙΝΑΚΕΣ

Σάρωση πίνακα

Για τη σάρωση μονοδιάστατου πίνακα απαιτείται μία δομή επανάληψης **ΓΙΑ**, ενώ για τη σάρωση δισδιάστατου πίνακα χρειάζεται και $2^η$ **ΓΙΑ** εμφωλευμένη στην $1^η$.
Εάν η μεταβλητή της εξωτερικής **ΓΙΑ** είναι ο $1^{ος}$ δείκτης του πίνακα έχω σάρωση κατά γραμμές, διαφορετικά κατά στήλες.

Να γίνει ανάγνωση των τιμών του πίνακα ΤΙΜΕΣ[100]

```
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 100
    ΔΙΑΒΑΣΕ ΤΙΜΕΣ[i]
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Να γίνει ανάγνωση α) κατά γραμμές και β) κατά στήλες των τιμών του πίνακα ΤΙΜΕΣ[10, 20]

```
! α) Γραμμές

ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 20
        ΔΙΑΒΑΣΕ ΤΙΜΕΣ[i, j]
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

```
! β) Στήλες

ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 20
    ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10
        ΔΙΑΒΑΣΕ ΤΙΜΕΣ[i, j]
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Εισαγωγή τιμών σε πίνακα

Να διαβαστούν τα ονόματα 16 ομάδων και να καταχωρηθούν στο μονοδιάστατο πίνακα ΟΝΟΜΑ καθώς και οι βαθμολογίες τους σε 30 αγώνες, οι οποίες να καταχωρηθούν στον δισδιάστατο πίνακα ΒΑΘΜΟΙ, με έλεγχο εγκυρότητας για τις βαθμολογίες, ώστε να μην εισάγονται αρνητικοί αριθμοί.

```
! Χωριστά

ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 16
    ΔΙΑΒΑΣΕ ΟΝΟΜΑ[i]
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 16
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 30
        ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
            ΔΙΑΒΑΣΕ ΒΑΘΜΟΙ[i, j]
        ΜΕΧΡΙΣ_ΟΤΟΥ ΒΑΘΜΟΙ[i, j]>=0
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

```
! Μαζί

ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 16
    ΔΙΑΒΑΣΕ ΟΝΟΜΑ[i]
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 30
        ΑΡΧΗ_ΕΠΑΝΑΛΗΨΗΣ
            ΔΙΑΒΑΣΕ ΒΑΘΜΟΙ[i, j]
        ΜΕΧΡΙΣ_ΟΤΟΥ ΒΑΘΜΟΙ[i, j]>=0
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Διαχείριση ψηφίων ακεραίου

Αν **x** ο ελεγχόμενος αριθμός, τότε με επαναληπτική χρήση των εντολών **x1 ← x MOD 10** και **x2 ← x DIV 10**, εκχωρούμε σε πίνακα Α[ν] τα ν ψηφία του (από το τελευταίο προς το πρώτο) τα οποία στη συνέχεια διαχειριζόμαστε ανά περίπτωση.

Να διαβαστεί εξαψήφιος ακέραιος και να σχηματιστεί ο αντίστοιχος ακέραιος με αντεστραμμένη τη σειρά των ψηφίων του (από το τελευταίο προς το πρώτο).

```
ΔΙΑΒΑΣΕ x                     ! έστω ότι δίνεται ο εξαψήφιος 135790
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 6
    A[i] ← x MOD 10           ! A[1] = 0 κ.ο.κ.
    x ← x DIV 10              ! x = 13579 κ.ο.κ.
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ              ! A[1]=0, A[2]=9, A[3]=7, A[4]=5, A[5]=3, A[6]=1
x ← 0
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 6
    x ← x + A[i] * 10 ^ (6 – i)   ! x = 0 + A[1] * 10 ^ 5 = 0 κ.ο.κ.
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ x                       ! x = 97531
```

Τετραγωνικοί Πίνακες

Εάν ο δείκτης i αφορά τις γραμμές και ο δείκτης j τις στήλες ενός τετραγωνικού πίνακα Α[Ν, Ν], τότε:

- Τα στοιχεία της κυρίας διαγωνίου καθορίζονται από τη σχέση $i = j$.
- Τα στοιχεία που βρίσκονται κάτω από την κύρια διαγώνιο καθορίζονται από τη σχέση $i > j$.
- Τα στοιχεία που βρίσκονται πάνω από την κύρια διαγώνιο καθορίζονται από τη σχέση $i < j$.
- Τα στοιχεία της δευτερεύουσας διαγωνίου καθορίζονται από τη σχέση $i + j = N + 1$.

Σε έναν πίνακα Α[10, 10] να τοποθετηθεί ο χαρακτήρας 'X' στην κύρια διαγώνιο, ο χαρακτήρας 'O' στη δευτερεύουσα διαγώνιο, ο χαρακτήρας '–' στα υπόλοιπα στοιχεία κάτω από την κύρια διαγώνιο και ο χαρακτήρας '+' στα υπόλοιπα στοιχεία πάνω από την κύρια διαγώνιο

```
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ 10                ! σάρωση γραμμών
   ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ 10             ! σάρωση στηλών
      ΑΝ i = j ΤΟΤΕ                 ! κύρια διαγώνιος
         ΓΡΑΨΕ 'X'
      ΑΛΛΙΩΣ_ΑΝ i + j = 11 ΤΟΤΕ     ! δευτερεύουσα διαγώνιος
         ΓΡΑΨΕ 'O'
      ΑΛΛΙΩΣ_ΑΝ i > j ΤΟΤΕ          ! κάτω από την κύρια διαγώνιο
         ΓΡΑΨΕ '–'
      ΑΛΛΙΩΣ                        ! πάνω από την κύρια διαγώνιο
         ΓΡΑΨΕ '+'
      ΤΕΛΟΣ_ΑΝ
   ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Καταμέτρηση υπό συνθήκη - Ποσοστά

Η αρχικοποίηση του μετρητή C ← 0 γίνεται πριν τις **ΓΙΑ**. Εάν το ποσοστό αναφέρεται σε ολόκληρο τον πίνακα, διαιρούμε με το πλήθος των στοιχείων του πίνακα (το μέγεθός του εάν πρόκειται για μονοδιάστατο ή το γινόμενο των διαστάσεών του εάν πρόκειται για δισδιάστατο), μετά τις **ΓΙΑ**.

Να εκτυπωθούν το πλήθος και το ποσοστό των στοιχείων ενός μονοδιάστατου πίνακα αριθμών A[N], που είναι θετικοί αριθμοί.

```
C ← 0                              ! αρχικοποίηση του μετρητή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ N
   ΑΝ A[i] > 0 ΤΟΤΕ
      C ← C + 1                    ! υπολογισμός πλήθους
   ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ποσοστό ← (C / N) * 100            ! υπολογισμός ποσοστού
ΓΡΑΨΕ C, ποσοστό
```

Να εκτυπωθούν το πλήθος και το ποσοστό των στοιχείων ενός δισδιάστατου πίνακα ακεραίων A[M, N], που είναι άρτιοι αριθμοί.

```
C ← 0                              ! αρχικοποίηση του μετρητή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ M                ! σάρωση γραμμών
   ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ N             ! σάρωση στηλών
      ΑΝ A[i, j] mod 2 = 0 ΤΟΤΕ
         C ← C + 1                 ! υπολογισμός πλήθους
      ΤΕΛΟΣ_ΑΝ
   ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ποσοστό ← (C / N) * 100            ! υπολογισμός ποσοστού
ΓΡΑΨΕ C, ποσοστό
```

Εάν ζητάμε το ποσοστό πάνω σε άλλη ποσότητα και όχι σε ολόκληρο τον πίνακα, πρέπει να ελέγχουμε εάν ο παρονομαστής είναι διάφορος του μηδενός.

Να εκτυπωθούν το πλήθος στοιχείων ενός μονοδιάστατου πίνακα ακεραίων A[N], που είναι θετικοί αριθμοί, το πλήθος αυτών που είναι άρτιοι και το ποσοστό των άρτιων στο σύνολο των θετικών.

```
C1 ← 0                             ! αρχικοποίηση του μετρητή των θετικών
C2 ← 0                             ! αρχικοποίηση του μετρητή των άρτιων
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ N
   ΑΝ A[i] > 0 ΤΟΤΕ
      C1 ← C1 + 1                  ! υπολογισμός πλήθους θετικών
   ΤΕΛΟΣ_ΑΝ
   ΑΝ A[i] mod 2 = 0 ΤΟΤΕ
      C2 ← C2 + 1                  ! υπολογισμός πλήθους θετικών
   ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ C1, C2
ΑΝ C1<>0 ΤΟΤΕ                            !εάν υπήρξαν θετικοί τότε…
   ποσοστό ← (C2 / C1) * 100             ! υπολογίζουμε το ποσοστό
   ΓΡΑΨΕ ποσοστό
ΤΕΛΟΣ_ΑΝ
```

Αθροίσματα - Μέσοι Όροι

➢ Ολόκληρου του πίνακα

Η αρχικοποίηση του αθροιστή S ← 0 γίνεται πριν τις **ΓΙΑ**. Για τον υπολογισμό του μέσου όρου διαιρούμε με το πλήθος των στοιχείων του πίνακα (το μέγεθός του εάν πρόκειται για μονοδιάστατο ή το γινόμενο των διαστάσεών του εάν πρόκειται για δισδιάστατο), μετά τις **ΓΙΑ**.

Να εκτυπωθούν το άθροισμα και ο μέσος όρος των στοιχείων ενός μονοδιάστατου πίνακα Α[Ν].

```
S ← 0                          ! αρχικοποίηση του αθροιστή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ N
    S ← S + A[i]               ! υπολογισμός αθροίσματος
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΜΟ ← S / N                     ! υπολογισμός μέσου όρου
ΓΡΑΨΕ S, ΜΟ
```

Να εκτυπωθούν το άθροισμα και ο μέσος όρος των στοιχείων ενός δισδιάστατου πίνακα Α[Μ, Ν].

```
S ← 0                          ! αρχικοποίηση του αθροιστή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ M            ! σάρωση γραμμών
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ N        ! σάρωση στηλών
        S ← S + A[i, j]        ! υπολογισμός αθροίσματος
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΜΟ ← S / (M * N)               ! υπολογισμός μέσου όρου
ΓΡΑΨΕ S, ΜΟ
```

➢ Κάθε γραμμής δισδιάστατου πίνακα

Η αρχικοποίηση του αθροιστή γίνεται ανάμεσα στις δύο **ΓΙΑ**, ενώ το αποτέλεσμα είναι διαθέσιμο ανάμεσα στα δύο **ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ** με τον μέσο όρο να υπολογίζεται ως το πηλίκο του αθροιστή προς το πλήθος των στηλών.

Να υπολογισθούν το άθροισμα και ο μέσος κάθε γραμμής ενός δισδιάστατου πίνακα Α[Μ, Ν] και να εκχωρηθούν στους μονοδιάστατους πίνακες ΑΘΡΟΙΣΜΑ[Μ] και ΜΕΣΟΣ[Μ] αντίστοιχα.

```
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ M            ! σάρωση γραμμών
    S ← 0                      ! αρχικοποίηση του αθροιστή
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ N        ! σάρωση στηλών
        S ← S + A[i, j]        ! υπολογισμός αθροίσματος γραμμής
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    ΜΟ ← S / N                 ! υπολογισμός μέσου όρου γραμμής
    ΑΘΡΟΙΣΜΑ[i] ← S
    ΜΕΣΟΣ[i] ← ΜΟ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

➢ Κάθε στήλης δισδιάστατου πίνακα

Η αρχικοποίηση του αθροιστή γίνεται ανάμεσα στις δύο **ΓΙΑ**, ενώ το αποτέλεσμα είναι διαθέσιμο ανάμεσα στα δύο **ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ** με τον μέσο όρο να υπολογίζεται ως το πηλίκο του αθροιστή προς το πλήθος των γραμμών.

Να υπολογισθούν το άθροισμα και ο μέσος κάθε στήλης ενός δισδιάστατου πίνακα Α[Μ, Ν] και να εκχωρηθούν στους μονοδιάστατους πίνακες ΑΘΡΟΙΣΜΑ[Μ] και ΜΕΣΟΣ[Μ] αντίστοιχα.

```
ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ Ν              ! σάρωση στηλών
    S ← 0                        ! αρχικοποίηση του αθροιστή
    ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ Μ          ! σάρωση γραμμών
        S ← S + A[i, j]          ! υπολογισμός αθροίσματος στήλης
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    ΜΟ ← S / M                   ! υπολογισμός μέσου όρου στήλης
    ΑΘΡΟΙΣΜΑ[j] ← S
    ΜΕΣΟΣ[j] ← ΜΟ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

➢ Συγκεκριμένης γραμμής δισδιάστατου πίνακα

Σαν να έχουμε μονοδιάστατο πίνακα με τον πρώτο δείκτη του πίνακα να είναι σταθερά ο δείκτης της συγκεκριμένης γραμμής π.χ. **κ** (δηλαδή αντί για i έχουμε κ).

Να εκτυπωθούν το άθροισμα και ο μέσος όρος της 3ης γραμμής ενός δισδιάστατου πίνακα Α[Μ, Ν]. (Μ>=3)

```
S ← 0                            ! αρχικοποίηση του αθροιστή
ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ Ν              ! σάρωση στηλών
    S ← S + A[3, j]              ! υπολογισμός αθροίσματος 3ης γραμμής
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΜΟ ← S / N                       ! υπολογισμός μέσου όρου 3ης γραμμής
ΓΡΑΨΕ S, ΜΟ
```

➢ Συγκεκριμένης στήλης δισδιάστατου πίνακα

Σαν να έχουμε μονοδιάστατο πίνακα με τον δεύτερο δείκτη του πίνακα να είναι σταθερά ο δείκτης της συγκεκριμένης στήλης π.χ. **λ** (δηλαδή αντί για j έχουμε λ).

Να εκτυπωθούν το άθροισμα και ο μέσος όρος της 4ης στήλης ενός δισδιάστατου πίνακα Α[Μ, Ν]. (Ν>=4)

```
S ← 0                            ! αρχικοποίηση του αθροιστή
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ Μ              ! σάρωση γραμμών
    S ← S + A[i, 4]              ! υπολογισμός αθροίσματος 4ης στήλης
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΜΟ ← S / M                       ! υπολογισμός μέσου όρου 4ης στήλης
ΓΡΑΨΕ S, ΜΟ
```

Μέγιστα - Ελάχιστα

➢ Ολόκληρου του πίνακα

Η αρχικοποίηση των **min** και **max** γίνεται με το A[1] για μονοδιάστατο πίνακα και το A[1,1] για δισδιάστατο πίνακα, πριν τις **ΓΙΑ**.

Να εκτυπωθεί η ΜΟΝΑΔΙΚΗ μέγιστη τιμή ενός μονοδιάστατου πίνακα A[N] και η ΘΕΣΗ της στον πίνακα.

```
max ← A[1]                      ! αρχικοποίηση μέσω υπάρχουσας τιμής
pos ← 1                         ! αρχικοποίηση θέσης
ΓΙΑ i ΑΠΟ 2 ΜΕΧΡΙ N             ! διαβάζουμε και τις υπόλοιπες τιμές
    ΑΝ A[i] > max ΤΟΤΕ          ! εάν η τρέχουσα τιμή είναι μεγαλύτερη
        max ← A[i]              ! τότε αυτή γίνεται το μέγιστο
        pos ← i                 ! και αυτή είναι τη θέση της
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ max, pos
```

ΘΕΣΕΙΣ ΕΛΑΧΙΣΤΟΥ. Να εκτυπωθεί η ελάχιστη τιμή ενός μονοδιάστατου πίνακα A[N] και οι ΘΕΣΕΙΣ του πίνακα, των οποίων οι τιμές είναι ίσες με τη ελάχιστη τιμή του.

```
min ← A[1]                      ! αρχικοποίηση μέσω υπάρχουσας τιμής
ΓΙΑ i ΑΠΟ 2 ΜΕΧΡΙ N             ! διαβάζουμε και τις υπόλοιπες τιμές
    ΑΝ A[i] < min ΤΟΤΕ          ! εάν η τρέχουσα τιμή είναι μικρότερη
        min ← A[i]              ! τότε αυτή γίνεται το ελάχιστο
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ min                       ! εκτύπωση της ελάχιστης τιμής
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ N             ! νέα σάρωση του πίνακα
    ΑΝ A[i] = min ΤΟΤΕ          ! εάν η τρέχουσα τιμή είναι ίση με το min
        ΓΡΑΨΕ i                 ! εκτυπώνεται η θέση της
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Να εκτυπωθεί η ΜΟΝΑΔΙΚΗ μέγιστη τιμή ενός δισδιάστατου πίνακα A[M, N] και η ΘΕΣΗ της στον πίνακα.

```
max ← A[1,1]                    ! αρχικοποίηση μέσω υπάρχουσας τιμής
grammi ← 1                      ! αρχικοποίηση γραμμής
stili ← 1                       ! αρχικοποίηση στήλης
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ M             ! σάρωση γραμμών
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ N         ! σάρωση στηλών
        ΑΝ A[i, j] > max ΤΟΤΕ   ! εάν η τρέχουσα τιμή είναι μεγαλύτερη
            max ← A[i, j]       ! τότε αυτή γίνεται το μέγιστο
            grammi ← i          ! και αυτή είναι τη γραμμή της
            stili ← j           ! και αυτή είναι τη στήλη της
        ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ max, grammi, stili
```

ΘΕΣΕΙΣ ΜΕΓΙΣΤΟΥ. Να εκτυπωθούν όλες οι θέσεις του δισδιάστατου πίνακα Α[Μ, Ν] που η τιμή τους ισούται με το μέγιστο του πίνακα.

```
max ← A[1,1]                     ! αρχικοποίηση μέσω υπάρχουσας τιμής
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ Μ              ! σάρωση γραμμών
   ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ Ν           ! σάρωση στηλών
      ΑΝ A[i, j] > max ΤΟΤΕ      ! εάν η τρέχουσα τιμή είναι μεγαλύτερη
         max ← A[i, j]           ! τότε αυτή γίνεται το μέγιστο
      ΤΕΛΟΣ_ΑΝ
   ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ

ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ Μ              ! σάρωση γραμμών
   ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ Ν           ! σάρωση στηλών
      ΑΝ A[i, j] = max ΤΟΤΕ      ! εάν η τρέχουσα τιμή είναι ίση με το max
         ΓΡΑΨΕ i, j              ! τότε γράψε γραμμή και στήλη
      ΤΕΛΟΣ_ΑΝ
   ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

- **Τμήματος του πίνακα**

Η αρχικοποίηση να πρέπει να γίνει είτε μέσω "ακραίας" τιμής, είτε μέσω του τελευταίου στοιχείου που ανήκει στο τμήμα του πίνακα που μας ενδιαφέρει, αφού πρώτα το εντοπίσουμε.

Δίνονται δύο ΠΑΡΑΛΛΗΛΟΙ πίνακες, ο Κ[Ν] που περιέχει τις κατευθύνσεις Ν μαθητών (με δυνατές τιμές ΘΕΤ, ΘΕΩ, ΤΕΧ) και ο Β[Ν] που περιέχει τους αντίστοιχους βαθμούς (στην κλίμακα 0 – 20). Να βρεθεί ο μέγιστος βαθμός της τεχνολογικής (ΤΕΧ) κατεύθυνσης.

```
max ← –1                         ! αρχικοποίηση μέσω ακραίας τιμής
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ Ν
   ΑΝ Κ[i] = 'ΤΕΧ' ΚΑΙ Β[i] > max ΤΟΤΕ
      max ← Β[i]
   ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ max
                      - - ή - -
pos ← 0                          ! αρχικοποίηση θέσης …
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ Ν
   ΑΝ Κ[i] = 'ΤΕΧ' ΤΟΤΕ
      pos ← i                    !...τελευταίου μαθητή τεχνολογικής στον πίνακα
   ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΑΝ pos <> 0 ΤΟΤΕ                 ! εάν υπάρχει μαθητής της τεχνολογικής
   max ← Β[pos]                  ! αρχικοποίηση μέσω υπάρχουσας τιμής
   ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ Ν           ! θα μπορούσε να είναι και ΜΕΧΡΙ pos – 1
      ΑΝ Κ[i] = 'ΤΕΧ' ΚΑΙ Β[i] > max ΤΟΤΕ
         max ← Β[i]
      ΤΕΛΟΣ_ΑΝ
   ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
   ΓΡΑΨΕ max
ΤΕΛΟΣ_ΑΝ
```

➢ Κάθε γραμμής δισδιάστατου πίνακα

Η αρχικοποίηση γίνεται ανάμεσα στις δύο **ΓΙΑ** με την τιμή A[i, 1] (δηλ. με το πρώτο στοιχείο κάθε γραμμής), ενώ το αποτέλεσμα είναι διαθέσιμο ανάμεσα στα δύο **ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ**.

Να υπολογισθεί η μέγιστη τιμή κάθε γραμμής ενός δισδιάστατου πίνακα A[M, N] και να εκχωρηθεί σε μονοδιάστατο πίνακα B[M].

```
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ M             ! σάρωση γραμμών
    max ← A[i, 1]               ! αρχικοποίηση max τρέχουσας γραμμής
    ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ N         ! σάρωση στηλών
        ΑΝ A[i, j] > max ΤΟΤΕ   ! εάν η τρέχουσα τιμή είναι μεγαλύτερη
            max ← A[i, j]       ! τότε αυτή γίνεται το μέγιστο
        ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    B[i] ← max                  ! εκχώρηση μέγιστου τρέχουσας γραμμής
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

➢ Κάθε στήλης δισδιάστατου πίνακα

Η αρχικοποίηση γίνεται ανάμεσα στις δύο **ΓΙΑ** με την τιμή A[1, j] (δηλ. με το πρώτο στοιχείο κάθε στήλης), ενώ το αποτέλεσμα είναι διαθέσιμο ανάμεσα στα δύο **ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ**.

Να υπολογισθεί η ελάχιστη τιμή κάθε στήλης ενός δισδιάστατου πίνακα A[M, N] και να εκχωρηθεί σε μονοδιάστατο πίνακα B[N].

```
ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ N             ! σάρωση στηλών
    min ← A[1, j]               ! αρχικοποίηση min τρέχουσας στήλης
    ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ M         ! σάρωση γραμμών
        ΑΝ A[i, j] < min ΤΟΤΕ   ! εάν η τρέχουσα τιμή είναι μικρότερη
            min ← A[i, j]       ! τότε αυτή γίνεται το ελάχιστο
        ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
    B[j] ← min                  ! εκχώρηση ελάχιστου τρέχουσας στήλης
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

➢ Συγκεκριμένης γραμμής δισδιάστατου πίνακα

Αρχικοποιούμε με το στοιχείο A[κ, 1], όπου **κ** η συγκεκριμένη γραμμή και στη συνέχεια με μία **ΓΙΑ** σαρώνουμε όλες τις στήλες της κ γραμμής (δηλαδή αντί για i έχουμε κ), σαν να επρόκειτο για μονοδιάστατο πίνακα.

Να εκτυπωθεί η μέγιστη τιμή της 3^{ης} γραμμής ενός δισδιάστατου πίνακα A[M, N]. (M>=3)

```
max ← A[3, 1]                   ! αρχικοποίηση max 3ης γραμμής
ΓΙΑ j ΑΠΟ 1 ΜΕΧΡΙ N             ! σάρωση στηλών
    ΑΝ A[3, j] > max ΤΟΤΕ       ! εάν η τρέχουσα τιμή είναι μεγαλύτερη
        max ← A[3, j]           ! τότε αυτή γίνεται το μέγιστο
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ max                       ! εκτύπωση μέγιστου 3ης γραμμής
```

➢ **Συγκεκριμένης στήλης δισδιάστατου πίνακα**

Αρχικοποιούμε με το στοιχείο A[1, λ], όπου **λ** η συγκεκριμένη στήλη και στη συνέχεια με μία **ΓΙΑ** σαρώνουμε όλες τις γραμμές της λ στήλης (δηλαδή αντί για j έχουμε λ), σαν να επρόκειτο για μονοδιάστατο πίνακα.

Να εκτυπωθεί η ελάχιστη τιμή της 4ης στήλης ενός δισδιάστατου πίνακα A[M, N]. (N>=4)

```
min ← A[1, 4]                       ! αρχικοποίηση min 4ης στήλης
ΓΙΑ i ΑΠΟ 1 ΜΕΧΡΙ M                 ! σάρωση γραμμών
    ΑΝ A[i, 4] < min ΤΟΤΕ           ! εάν η τρέχουσα τιμή είναι μικρότερη
        min ← A[i, 4]               ! τότε αυτή γίνεται το ελάχιστο
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ min                           ! εκτύπωση ελάχιστου 4ης στήλης
```

Σειριακή Αναζήτηση

Να αναζητεί στον πίνακα A[N] η τιμή της μεταβλητής key και να εκτυπωθεί η θέση του πίνακα στην οποία βρέθηκε.

```
i ← 1
pos ← 0
done ← ΨΕΥΔΗΣ
ΟΣΟ (done = ΨΕΥΔΗΣ) ΚΑΙ (i <= N) ΕΠΑΝΑΛΑΒΕ
    ΑΝ A[i] = key ΤΟΤΕ
        done ← ΑΛΗΘΗΣ
        pos ← i
    ΑΛΛΙΩΣ
        i ← i + 1
    ΤΕΛΟΣ_ΑΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΓΡΑΨΕ pos                           ! εάν είναι 0 σημαίνει ότι δεν βρέθηκε
```

Ταξινόμηση ευθείας ανταλλαγής (φυσαλίδας)

Να ταξινομηθεί σε αύξουσα σειρά ο πίνακας A[N].

```
ΓΙΑ i ΑΠΟ 2 ΜΕΧΡΙ N
    ΓΙΑ j ΑΠΟ N ΜΕΧΡΙ i ΜΕ ΒΗΜΑ -1
        ΑΝ A[j - 1] > A[j] ΤΟΤΕ         ! με "<" είναι φθίνουσα ταξινόμηση
            temp ← A[j - 1]             ! αντιμετάθεση
            A[j - 1] ← A[j]
            A[j] ← temp
        ΤΕΛΟΣ_ΑΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```

Συγχώνευση Ταξινομημένων Πινάκων

Να συγχωνευτούν οι σε αύξουσα σειρά ταξινομημένοι πίνακες ΠΙΝ1[Ν1] και ΠΙΝ2[Ν2] στον πίνακα ΠΙΝ[Ν1+Ν2].

```
δ1 ← 1                                      ! Δείκτης τρέχοντος στοιχείου του ΠΙΝ1
δ2 ← 1                                      ! Δείκτης τρέχοντος στοιχείου του ΠΙΝ2
δ ← 1                                       ! Δείκτης τρέχοντος στοιχείου του ΠΙΝ
ΟΣΟ (δ1 <= Ν1) ΚΑΙ (δ2 <= Ν2) ΕΠΑΝΑΛΑΒΕ     ! Όσο υπάρχουν στοιχεία και στους δύο πίνακες…
    ΑΝ (ΠΙΝ1[δ1] <= ΠΙΝ2[δ2]) ΤΟΤΕ          ! Εάν ο ΠΙΝ1 έχει το τρέχων μικρότερο στοιχείο…
        ΠΙΝ[δ] ← ΠΙΝ1[δ1]                   ! …το εκχωρούμε στον ΠΙΝ…
        δ1 ← δ1 + 1                         ! …και πάμε στο επόμενο στοιχείο του ΠΙΝ1
    ΑΛΛΙΩΣ                                  ! Εάν όμως ο ΠΙΝ2 έχει το τρέχων μικρότερο στοιχείο…
        ΠΙΝ[δ] ← ΠΙΝ2[δ2]                   ! …τότε εκχωρούμε αυτό στον ΠΙΝ…
        δ2 ← δ2 + 1                         ! …και πάμε στο επόμενο στοιχείο του ΠΙΝ2
    ΤΕΛΟΣ_ΑΝ
    δ ← δ + 1                               ! Ετοιμάζουμε τον επόμενο δείκτη του ΠΙΝ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ                            ! Μέχρι ένας από τους ΠΙΝ1 ή ΠΙΝ2 να "εξαντληθεί"
ΑΝ (δ1 <= Ν1) ΤΟΤΕ                          ! Εάν ο ΠΙΝ1 έχει ακόμα στοιχεία…
    ΓΙΑ i ΑΠΟ δ1 ΜΕΧΡΙ Ν1
        ΠΙΝ[δ]  ← ΠΙΝ1[i]                   ! …τα μεταφέρουμε στον ΠΙΝ
        δ ← δ + 1                           ! … και ετοιμάζουμε τον επόμενο δείκτη του ΠΙΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΑΛΛΙΩΣ                                      ! Αλλιώς, αν ο ΠΙΝ2 έχει ακόμα στοιχεία…
    ΓΙΑ i ΑΠΟ δ2 ΜΕΧΡΙ Ν2
        ΠΙΝ[δ]  ← ΠΙΝ2[i]                   ! …μεταφέρουμε αυτά στον ΠΙΝ
        δ ← δ + 1                           ! … και ετοιμάζουμε τον επόμενο δείκτη του ΠΙΝ
    ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
ΤΕΛΟΣ_ΑΝ
```

Επιλογή Στοιχείων από δύο Ταξινομημένους Πίνακες

Να βρεθούν τα τέσσερα μικρότερα στοιχεία των ταξινομημένων σε αύξουσα σειρά πινάκων Α[Ν] και Β[Μ] (Ν ≥ 4 , Μ ≥4) και να εκχωρηθούν στον πίνακα Γ[4].

```
δ1 ← 1                                      ! Δείκτης τρέχοντος στοιχείου του Α
δ2 ← 1                                      ! Δείκτης τρέχοντος στοιχείου του Β
δ ← 1                                       ! Δείκτης τρέχοντος στοιχείου του Γ
ΟΣΟ (δ <= 4) ΕΠΑΝΑΛΑΒΕ                      ! Μέχρι να βρεθούν τα 4 στοιχεία
    ΑΝ (Α[δ1] <= Β[δ2]) ΤΟΤΕ                ! Εάν ο Α έχει το τρέχων μικρότερο στοιχείο…
        Γ[δ] ← Α[δ1]                        ! …το εκχωρούμε στον Γ…
        δ1 ← δ1 + 1                         ! …και πάμε στο επόμενο στοιχείο του Α
    ΑΛΛΙΩΣ                                  ! Εάν όμως ο Β έχει το τρέχων μικρότερο στοιχείο…
        Γ[δ] ← Β[δ2]                        ! …τότε εκχωρούμε αυτό στον Γ…
        δ2 ← δ2 + 1                         ! …και πάμε στο επόμενο στοιχείο του Β
    ΤΕΛΟΣ_ΑΝ
    δ ← δ + 1                               ! Ετοιμάζουμε τον επόμενο δείκτη του Γ
ΤΕΛΟΣ_ΕΠΑΝΑΛΗΨΗΣ
```